COMMUNICATIONS IN APPLIED NUMERICAL METHODS, Vol. 2, 523–529 (1986)

AN EXPLICIT FORM OF THE PLASTIC MATRIX FOR A MOHR–COULOMB MATERIAL

D. V. GRIFFITHS AND S. M. WILLSON

Simon Engineering Laboratories, University of Manchester, Manchester, U.K.

SUMMARY

An explicit form of the plastic matrix for a Mohr–Coulomb material in two dimensions is presented. The derivation, which assumes elastic–perfectly plastic behaviour is obtained by direct differentiation of a non-invariant expression for the Mohr–Coulomb yield function. A FORTRAN 77 subroutine is also provided.

INTRODUCTION

Many algorithms for analysing elasto-plastic materials, such as tangent stiffness and constant stiffness (initial stress methods), require an elasto-plastic matrix which is a function of elastic properties and the assumed failure and potential functions.[1] For metal plasticity or undrained clays, the von Mises criterion is suitable, and this function leads to a rather simple plastic matrix which is easily formed explicitly.[2] For frictional materials, however, the plastic matrices have tended to be formed 'numerically' (see, for example, Reference 3). This was because the invariant form of the Mohr–Coulomb criterion leads to rather unwieldy expressions on differentiation.

The formulation presented here uses a non-invariant form of the Mohr–Coulomb function as the starting point. This enables the derivatives and the plastic matrix itself to be expressed in fairly simple algebraic form. Corners are dealt with in the usual way by the introduction of a smoothing function.

The small penalty for using this non-invariant approach for obtaining the Mohr–Coulomb derivatives is that care must be taken over the signs of some of the terms.

A brief review of the plastic matrix derivation follows.

GENERAL DERIVATION OF AN ELASTO-PLASTIC MATRIX

Consider an element of material in plane strain or axisymmetry with a stress state acting on it such that is lies on a yield surface. An increment of strain will generally contain both elastic and plastic components thus:

$$\mathbf{d}\boldsymbol{\epsilon} = \mathbf{d}\boldsymbol{\epsilon}^e + \mathbf{d}\boldsymbol{\epsilon}^p \tag{1}$$

Classical plasticity theory requires that plastic strain increments should occur normal to the yield surface F, hence

$$\mathbf{d}\boldsymbol{\epsilon}^p = \lambda \frac{\partial F}{\partial \boldsymbol{\sigma}} \tag{2}$$

where λ is a scalar multiplier.

For frictional materials, however, non-associated flow rules must be used to avoid excessive dilation. In such cases, strain increments occur normal to a potential surface Q where $Q \neq F$, hence

$$\mathbf{d}\boldsymbol{\epsilon}^p = \lambda \frac{\partial Q}{\partial \boldsymbol{\sigma}} \tag{3}$$

0748–8025/86/050523–07$05.00

Received March 1986

Assuming stress changes are generated by elastic strain component only, then

$$\mathbf{d\sigma} = \mathbf{D}^e\left(\mathbf{d\epsilon} - \lambda \frac{\partial Q}{\partial \boldsymbol{\sigma}}\right) \tag{4}$$

For elements already on the yield surface and in the absence of hardening or unloading, subsequent stress increments may shift the stress state to a different position on the surface, but not off it. Hence

$$\frac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{d\sigma} = 0 \tag{5}$$

$$\therefore \quad \frac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\left(\mathbf{D}^e\,\mathbf{d\epsilon} - \lambda\,\mathbf{D}^e\frac{\partial Q}{\partial \boldsymbol{\sigma}}\right) = 0 \tag{6}$$

$$\therefore \lambda = \frac{\dfrac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{D}^e\,\mathbf{d\epsilon}}{\dfrac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{D}^e\dfrac{\partial Q}{\partial \boldsymbol{\sigma}}} = \text{scalar} \tag{7}$$

$$\therefore \mathbf{d\sigma} = \mathbf{D}^e\,\mathbf{d\epsilon} - \left(\frac{\dfrac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{D}^e\,\mathbf{d\epsilon}}{\dfrac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{D}^e\dfrac{\partial Q}{\partial \boldsymbol{\sigma}}}\right)\mathbf{D}^e\frac{\partial Q}{\partial \boldsymbol{\sigma}} \tag{8}$$

$$= \mathbf{D}^e\mathbf{d\epsilon} - \frac{\mathbf{D}^e\dfrac{\partial Q}{\partial \boldsymbol{\sigma}}\dfrac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{D}^e}{\dfrac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{D}^e\dfrac{\partial Q}{\partial \boldsymbol{\sigma}}}\mathbf{d\epsilon} \tag{9}$$

hence

$$\mathbf{d\sigma} = (\mathbf{D}^e - \mathbf{D}^p)\,\mathbf{d\epsilon} \tag{10}$$

where

$$\mathbf{D}^p = \frac{\mathbf{D}^e\dfrac{\partial Q}{\partial \boldsymbol{\sigma}}\dfrac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{D}^e}{\dfrac{\partial F^{\mathrm{T}}}{\partial \boldsymbol{\sigma}}\mathbf{D}^e\dfrac{\partial Q}{\partial \boldsymbol{\sigma}}} = \text{plastic matrix} \tag{11}$$

PLASTIC MATRIX FOR A MOHR–COULOMB MATERIAL

The yield function can be expressed in many different forms (the invariant form is given in Appendix II), but should have units of stress and preferably the property that it equals zero if a given stress state lies on the yield surface.

For a 'compression is −ve' sign convention, the Mohr–Coulomb yield function may be expressed thus:

$$F = \sqrt{[(\sigma_x - \sigma_y)^2 + 4\tau_{xy}{}^2]} + (\sigma_x + \sigma_y)\sin\phi - 2c\cos\phi \tag{12}$$

In its simplest form, the plastic potential is algebraically similar to equation (12) except with the dilation angle ψ replacing the friction angle ϕ, hence:

$$Q = \sqrt{[(\sigma_x - \sigma_y)^2 + 4\tau_{xy}{}^2]} + (\sigma_x + \sigma_y)\sin\psi - 2c\cos\psi \tag{13}$$

Defining

$$\alpha = \arctan\left|\frac{\sigma_x - \sigma_y}{2\,\tau_{xy}}\right| \tag{14}$$

and putting

$$k_1 = 1 \text{ if } |\sigma_y| \geqslant |\sigma_x| \text{ , } k_1 = -1 \text{ if } |\sigma_x| > |\sigma_y| \text{ ,}$$
$$k_2 = 1 \text{ if } \tau_{xy} \geqslant 0 \text{ and } k_2 = -1 \text{ if } \tau_{xy} < 0$$

then

$$\frac{\partial F}{\partial \boldsymbol{\sigma}} = \begin{Bmatrix} \sin\phi + k_1 \sin\alpha \\ \sin\phi - k_1 \sin\alpha \\ 2\,k_2 \cos\alpha \\ 0 \end{Bmatrix} \tag{15}$$

$$\frac{\partial Q}{\partial \boldsymbol{\sigma}} = \begin{Bmatrix} \sin\psi + k_1 \sin\alpha \\ \sin\psi - k_1 \sin\alpha \\ 2\,k_2 \cos\alpha \\ 0 \end{Bmatrix} \tag{16}$$

where

$$\boldsymbol{\sigma} = \begin{Bmatrix} \sigma_x \\ \sigma_y \\ \tau_{xy} \\ \sigma_z \end{Bmatrix} \text{ in plane strain or } \begin{Bmatrix} \sigma_r \\ \sigma_z \\ \tau_{rz} \\ \sigma_t \end{Bmatrix} \text{ in axisymmetry} \tag{17}$$

Defining the elastic matrix for use in plane strain or axisymmetry as

$$\mathbf{D}^e = \frac{E}{(1+\nu)(1-2\nu)} \begin{bmatrix} 1-\nu & \nu & 0 & \nu \\ \nu & 1-\nu & 0 & \nu \\ 0 & 0 & \dfrac{1-2\nu}{2} & 0 \\ \nu & \nu & 0 & 1-\nu \end{bmatrix} \tag{18}$$

the plastic matrix may now be obtained explicitly by substitution into equation (11) to give

$$\mathbf{D}^p = \frac{E}{2(1+\nu)(1-2\nu)(1-2\nu+\sin\phi\sin\psi)}\mathbf{A} \tag{19}$$

where

$$\mathbf{A} = \begin{bmatrix} R_1C_1 & R_1C_2 & R_1C_3 & R_1C_4 \\ R_2C_1 & R_2C_2 & R_2C_3 & R_2C_4 \\ R_3C_1 & R_3C_2 & R_3C_3 & R_3C_4 \\ R_4C_1 & R_4C_2 & R_4C_3 & R_4C_4 \end{bmatrix} \tag{20}$$

and

$$\begin{aligned} C_1 &= \sin\phi + k_1(1-2\nu)\sin\alpha \\ C_2 &= \sin\phi - k_1(1-2\nu)\sin\alpha \\ C_3 &= k_2(1-2\nu)\cos\alpha \\ C_4 &= 2\nu\sin\phi \\ R_1 &= \sin\psi + k_1(1-2\nu)\sin\alpha \\ R_2 &= \sin\psi - k_1(1-2\nu)\sin\alpha \\ R_3 &= k_2(1-2\nu)\cos\alpha \\ R_4 &= 2\nu\sin\psi \end{aligned} \tag{21}$$

Note that $\mathbf{A}$ is symmetrical if $\phi = \psi$ (associated).

CORNERS ON THE YIELD AND POTENTIAL SURFACES

As can be seen from Figure 1, the derivatives of F and Q are indeterminate at the vertices of the



$$F_{MC} = \sigma_m \sin\phi + \sqrt{J_2}\left(\cos\theta - \frac{\sin\theta \sin\phi}{\sqrt{3}}\right) - c\cos\phi = 0$$

$$F_{TC} = \sigma_m \sin\phi + \frac{\sqrt{J_2}}{2}\left(\sqrt{3} - \frac{\sin\phi}{\sqrt{3}}\right) - c\cos\phi = 0$$

$$F_{TE} = \sigma_m \sin\phi + \frac{\sqrt{J_2}}{2}\left(\sqrt{3} + \frac{\sin\phi}{\sqrt{3}}\right) - c\cos\phi = 0$$

Figure 1. Surfaces used at corners of Mohr–Coulomb

hexagonal surface. These vertices correspond to triaxial stress states which occur when the angular invariant

$$\begin{aligned} &\theta = 30°, \quad \text{triaxial compression, } |\sigma_1| \geqslant |\sigma_2| = |\sigma_3| \\ &\text{or} \\ &\theta = -30°, \quad \text{triaxial extension, } |\sigma_1| = |\sigma_2| \geqslant |\sigma_3| \end{aligned} \tag{22}$$

If the above stress states should occur, the plastic matrix of equation (19) is no longer valid and an alternative approach must be adopted.

A rounding of the corners of Mohr–Coulomb (see, for example, Reference 4) has been used here. This means that if the angular invariant gets sufficiently close to $\pm 30°$, the plastic matrix is formulated according to either the triaxial compression or the triaxial extension cone. These are obtained by substituting $\theta = 30°$ or $-30°$, respectively, into the expression for Mohr–Coulomb in terms of invariants (Figure 1).

The plastic matrices for rounding the corners of Mohr–Coulomb are obtained, as before, from equation (11) where

$$F = \sigma_m \sin\phi + \frac{\sqrt{J_2}}{2}\left(\sqrt{3} \pm \frac{\sin\phi}{\sqrt{3}}\right) - c\cos\phi \tag{23}$$

and

$$Q = \sigma_m \sin\psi + \frac{\sqrt{J_2}}{2}\left(\sqrt{3} \pm \frac{\sin\psi}{\sqrt{3}}\right) - c\cos\psi \tag{24}$$

Using the chain rule where

$$\frac{\partial F}{\partial \sigma_x} = \frac{\partial F}{\partial \sigma_m}\frac{\partial \sigma_m}{\partial \sigma_x} + \frac{\partial F}{\partial J_2}\frac{\partial J_2}{\partial \sigma_x}, \text{ etc.} \tag{25}$$

then

$$\frac{\partial F}{\partial \boldsymbol{\sigma}} = \left\{ \begin{array}{c} \frac{\sin\phi}{3} + \frac{s_x}{4}\sqrt{\left(\frac{3}{J_2}\right)}\left(1 \pm \frac{\sin\phi}{3}\right) \\ \frac{\sin\phi}{3} + \frac{s_y}{4}\sqrt{\left(\frac{3}{J_2}\right)}\left(1 \pm \frac{\sin\phi}{3}\right) \\ \frac{\tau_{xy}}{2}\sqrt{\left(\frac{3}{J_2}\right)}\left(1 \pm \frac{\sin\phi}{3}\right) \\ \frac{\sin\phi}{3} + \frac{s_z}{4}\sqrt{\left(\frac{3}{J_2}\right)}\left(1 \pm \frac{\sin\phi}{3}\right) \end{array} \right\} \tag{26}$$

As before, $\partial Q/\partial \boldsymbol{\sigma}$ is identical to $\partial F/\partial \boldsymbol{\sigma}$ except with ψ substituted for ϕ. The plastic matrix may now be obtained explicitly to give

$$\mathbf{D}^p = \frac{E}{(1+\nu)(1-2\nu)\{K_\phi \sin\psi + 2C_\phi C_\psi J_2 (1-2\nu)\}}\mathbf{A} \tag{27}$$

where $\mathbf{A}$ is defined by equation (20) but with columns and rows given as follows:

$$\begin{aligned} C_1 &= K_\phi + C_\phi\{(1-\nu)s_x + \nu(s_y + s_z)\} \\ C_2 &= K_\phi + C_\phi\{(1-\nu)s_y + \nu(s_z + s_x)\} \\ C_3 &= C_\phi(1-2\nu)\tau_{xy} \\ C_4 &= K_\phi + C_\phi\{(1-\nu)s_z + \nu(s_x + s_y)\} \\ R_1 &= K_\psi + C_\psi\{(1-\nu)s_x + \nu(s_y + s_z)\} \\ R_2 &= K_\psi + C_\psi\{(1-\nu)s_y + \nu(s_z + s_x)\} \\ R_3 &= C_\psi(1-2\nu)\tau_{xy} \\ R_4 &= K_\psi + C_\psi\{(1-\nu)s_z + \nu(s_x + s_y)\} \end{aligned} \tag{28}$$

where

$$K_\phi = \frac{\sin\phi}{3}(1+\nu), \quad K_\psi = \frac{\sin\psi}{3}(1+\nu)$$

$$C_\phi = \frac{1}{4}\sqrt{\left(\frac{3}{J_2}\right)}\left(1 \pm \frac{\sin\phi}{3}\right) \text{ and } C_\psi = \frac{1}{4}\sqrt{\left(\frac{3}{J_2}\right)}\left(1 \pm \frac{\sin\psi}{3}\right) \tag{29}$$

The criterion for using the 'rounded' version of the plastic matrix is quite arbitrary, but in the present work the 'rounded' version is used if

$$|\sin\theta| \geqslant 0{\cdot}49, \text{ or } |\theta| \geqslant 29{\cdot}34°$$

Having established that $|\sin\theta| \geqslant 0{\cdot}49$, the sign of θ must then be observed in order to determine the sign in the expressions for C_ϕ and C_ψ in equation (29). If $\theta \simeq 30°$ then the $-$ve sign is correct, but if $\theta \simeq -30°$ then the +ve sign should be used.

APPENDIX I: THE SUBROUTINE

The subroutine which forms the plastic matrix for a Mohr–Coulomb material taking account of corners, is listed below. Input to the subroutine involves the following values:

PHI = The friction angle ϕ (in degrees).
PSI = The dilation angle ψ (in degrees).
E = Young's modulus E.
V = Poisson's ratio ν.
SX = The current stresses σ_x, σ_y, σ_z and τ_{xy}
SY for plane strain or
SZ σ_r, σ_z, σ_t and τ_{rz} for axisymmetry.
TXY

Output from the subroutine:

PL = The 4 × 4 plastic matrix $\mathbf{D}^p$.

```
      SUBROUTINE MOCOPL(PHI,PSI,E,V,SX,SY,TXY,SZ,PL)
C
C     THIS SUBROUTINE FORMS THE PLASTIC STRESS/STRAIN MATRIX
C     FOR A MOHR-COULOMB MATERIAL (PHI,PSI IN DEGREES)
C
      REAL ROW (4),COL(4),PL(4,4)
      PI=4.* ATAN(1.)
      PHIR=PHI*PI/180.
      PSIR=PSI*PI/180.
      SNPH=SIN(PHIR)
      SNPS=SIN(PSIR)
      SQ3=SQRT(3.)
      CC=1.-2.*V
      DX=(2.*SX-SY-SZ)/3.
      DY=(2.*SY-SZ-SX)/3.
      DZ=(2.*SZ-SX-SY)/3.
      D2=SQRT(-DX*DY-DY*DZ-DZ*DX+TXY*TXY)
      D3=DX*DY*DZ-DZ*TXY*TXY
      TH=-3.*SQ3*D3/(2.*SQRT(D2)**3)
      IF(TH.GT.1.)TH=1.
      IF(TH.LT.-1.)TH=-1.
      TH=ASIN(TH)/3.
      SNTH=SIN(TH)
      IF(ABS(SNTH).GT..49)THEN
      SIG=-1.
      IF(SNTH.LT.0.)SIG=1.
      RPH=SNPH*(1.+V)/3.
      RPS=SNPS*(1.+V)/3.
      CPS=.25*SQ3/D2*(1.+SIG*SNPS/3.)
      CPH=.25*SQ3/D2*(1.+SIG*SNPH/3.)
      COL(1)=RPH+CPH*((1.-V)*DX+V*(DY+DZ))
      COL(2)=RPH+CPH*((1.-V)*DY+V*(DZ+DX))
      COL(3)=CPH*CC*TXY
      COL(4)=RPH+CPH*((1.-V)*DZ+V*(DX+DY))
      ROW(1)=RPS+CPS*((1.-V)*DX+V*(DY+DZ))
      ROW(2)=RPS+CPS*((1.-V)*DY+V*(DZ+DX))
      ROW(3)=CPS*CC*TXY
      ROW(4)=RPS+CPS*((1.-V)*DZ+V*(DX+DY))
      EE=E/((1.+V)*CC*(RPH*SNPS+2.*CPH*CPS*D2*D2*CC))
      ELSE
      ALP=ATAN(ABS((SX-SY)/(2.*TXY)))
      CA=COS(ALP)
      SA=SIN(ALP)
      DD=CC*SA
      S1=1.
      S2=1.
      IF((SX-SY).LT..0)S1=-1
      IF(TXY.LT..0)S2=-1.
      COL(1)=SNPH+S1*DD
      COL(2)=SNPH-S1*DD
      COL(3)=S2*CC*CA
      COL(4)=2.*V*SNPH
      ROW(1)=SNPS+S1*DD
      ROW(2)=SNPS-S1*DD
      ROW(3)=S2*CC*CA
      ROW(4)=2.*V*SNPS
      EE=E/(2.*(1.+V)*CC*(SNPH*SNPS+CC))
      END IF
      DO 1 I=1,4
      DO 1 J=1,4
    1 PL(I,J)=EE*ROW(I)*COL(J)
      RETURN
      END
```

APPENDIX II: INVARIANTS

$$\sigma_m = (\sigma_x + \sigma_y + \sigma_z)/3$$
$$J_2 = - s_x s_y - s_y s_z - s_z s_x + \tau_{xy}^2$$
$$J_3 = s_x s_y s_z - s_z \tau_{xy}^2$$

Invariants

$$\theta = \frac{1}{3} \arcsin \left(\frac{-3\sqrt{3}J_3}{2J_2^{3/2}} \right)$$

where

$$s_x = (2\,\sigma_x - \sigma_y - \sigma_z)/3, \text{ etc.}$$

Invariant form of Mohr–Coulomb:

$$F = \sigma_m \sin\phi + \sqrt{J_2}\left(\cos\theta - \frac{\sin\theta \sin\phi}{\sqrt{3}}\right) - c\cos\phi = 0$$

REFERENCES

1. O. C. Zienkiewicz, *The Finite Element Method*, McGraw-Hill, New York, pp. 462–471, 1977.
2. Y. Yamada, N. Yoshimura, and T. Sakurai, 'Plastic stress–strain matrix and its application for the solution of elastic–plastic problems by the finite element method', *Int. J. Mech. Sci.*, **10**, 343–354 (1968).
3. I. M. Smith, *Programming the Finite Element Method*, Wiley, New York, 1982.
4. O. C. Zienkiewicz, and C. Humpheson, 'Viscoplasticity: a generalised model for description of soil behaviour', in *Numerical Methods in Geotechnical Engineering* (Eds. C. S. Desai and J. T. Christian), McGraw-Hill, New York, pp. 116–147, 1977.